INTRASTAT v IS Bílý Motýl

Odkazy:

Hlavní stránka Celní správy k Intrastatu: http://www.cs.mfcr.cz/CmsGrc/Obchod-se-zbozim/Intrastat/

Aktuální číselníky pro Intrastat: http://dw.czso.cz/pls/metis/!cisprocel.cis

Vyhledávání v TARICu: http://ec.europa.eu/taxation_customs/dds/cgi-bin/tarchap?Lang=CS

1. Nutné minimum

1.1. Obecný výběr „Intrastat“ nad událostmi obchodu 

Obecný výběr by měl vybírat následující události obchodu:

· Jedná se o zboží (mohou být různé druhy artiklů, dle konkrétní implementace)
· Jedná se o faktury nebo dobropisy
· Datum události (dodání) je v zadaném roce a měsíci (nejlepší je propojení faktur s událostí skladu a datum brát z událostí skladu, viz druhý příklad obecného výběru)
· Adresát, respektive příjemce je z EU

!!! Faktury přijaté a dobropisy přijaté za zboží (skladované artikly) je potřeba zadávat položkově !!!	

Příklady ov:

YEAR(OBUDAL.ob02datdod)=[ROK] and MONTH(OBUDAL.ob02datdod)=[MESIC] and 
(
(
LEFT(SPPRC.sp07idtypd,4) in ('OBVF','OBVD') and 
exists(select * from %%%.ADADR ADR, %%%.ADSPOJ ADJ where ADR.ad01id=OBUDAL.ad01id1 and ADR.ad02cisadr=ADJ.ad02cisadr and  ADJ.ad01id=OBUDAL.ad01id1 and  ADJ.ad11cisspo=OBUDAL.ad11cissp1 and ADR.dp04tuzzah=3) 
)
or 
(
LEFT(SPPRC.sp07idtypd,4) in ('OBPF','OBPD') and 
exists(select * from %%%.ADADR ADR, %%%.ADSPOJ ADJ where ADR.ad01id=OBUDAL.ad01id and ADR.ad02cisadr=ADJ.ad02cisadr and  ADJ.ad01id=OBUDAL.ad01id and  ADJ.ad11cisspo=OBUDAL.ad11cisspo and ADR.dp04tuzzah=3) 
)
)
and 
LEFT(OBUDAL.sp61id,1) in ('Z')

---------------------

LEFT(SPPRC.sp07idtypd,4) in ('OBVF','OBPF','OBVD','OBPD') and YEAR(OBUDAL.ob02dat)=[ROK] and MONTH(OBUDAL.ob02dat)=[MESIC] and exists(select * from %%%.ADADR AD where AD.ad01id=OBUDAL.ad01id and AD.dp04tuzzah=3) and LEFT(OBUDAL.sp61id,2) in ('AS','S1','S2')

---------------------
LEFT(SPPRC.sp07idtypd,4) in ('OBVF','OBPF','DEVF','OBVD','OBPD') and exists(select * from %%%.ADADR AD where AD.ad01id=OBUDAL.ad01id1 and AD.dp04tuzzah=3) and OBUDAL.sp61id in ('MAT','POL','VYR','ZBO') and 
exists(select * from %%%.MTUDA MT, %%%.SPPRC SPMT where 
SPPRC.sp33hid=SPMT.sp33hid and 
SPMT.sp33cisd=MT.sp33cisd and 
SPMT.sp33cisdr=MT.sp33cisdr and 
SPMT.sp33cisdpm=MT.sp33cisdpm and 
SPMT.sp33storno=1 and 
SPMT.sp07idtypd in ('SKPP01','SKPP02','SKPR02','SKVP03','SKVP05','SKVR02') and 
YEAR(MT.mt02dat)=[ROK] and MONTH(MT.mt02dat)=[MESIC] and 
MT.sp61id=OBUDAL.sp61id and MT.sp60id=OBUDAL.sp60id)


1.2. Klasifikace událostí obchodu

Číselník klasifikací by měl obsahovat kombinace, které daná firma využívá. 

Řetězec se skládá z X.Y.ZZZ.WW, kde 

X 		= povaha transakce
Y 		= druh dopravy
ZZZ	= dodací podmínky
WW 	= zvláštní pohyb

1.3. Validace pro vyplnění klasifikace

U konkrétních procesů by měla být v masce operace procesu operace obchodu validace na vyplnění klasifikace.

Příklad takové validace:

(not empty(###) and ALLTRIM(###)<>'-' and INLIST(sp61id,'MAT','POL','VYR') and dp04tuzzah=3) or dp04tuzzah<>3 or !INLIST(sp61id,'MAT','POL','VYR')

1.4. Číselník Celní sazebník a Sazby celního sazebníku

Položky sazebníků lze naimportovat z aktuálních číselníků, viz odkazy.

Pokud firma používá kódy zboží, u kterých se vykazuje DMJ – Doplňková měrná jednotka, je potřeba založit v IS BM nové měrné jednotky pro Intrastat, například:

MTK - Metr čtvereční INTRASTAT
KGN – Kilogram INTRASTAT apod.

Číselník měrných jednotek je též obsažen v archívu číselníků, viz odkazy.

!!! Pokud je DMJ různá od „ZZZ“, pak je povinnost ji vykazovat a není pak nutné vykazovat hmotnost, pokud je DMJ „ZZZ“, je povinnost uvádět hmotnost. !!! 

1.5. Adresář INSTRASTATONLINE

V adresáři IS BM musí existovat adresář INTRASTATONLINE a v něm musí být soubor 7za.exe, což je freewarový zip, který umí dlouhé názvy souborů.

1.6. Měny a kursy Celní správy
Z důvodu častého vypisování dvou a více kursů měny Celním úřadem na vykazovaný měsíc došlo ke změně v záznamu kursů pro vykazování Intrastatu, nyní jsou kursy Celního úřadu ukládány v komponentě Oceňování v IS BM. Syntaxe měn je shodná na prvních dvou místech s reálnými měnami, na třetím místě je znak #. 
[image: ]

1.7. Evidování země odeslání, určení a země původu zboží

Dovoz – země odeslání je odvozena z adresáta (ad01id), země původu je odvozena z výrobního čísla nakupovaného zboží (detailní informace poskytne implementátor) nebo z volné karty Intrastat, která byla tímto atributem rozšířena
Vývoz – země určení je odvozena z příjemce (ad01id1), kraj původu je odvozen z výrobního čísla prodávaného zboží (detailní informace poskytne implementátor) nebo z volné karty Intrastat, která byla tímto atributem rozšířena
S touto změnou úzce souvisí nutnost evidování skladových karet s údaji o místě původu, tedy nutnost použití výrobního čísla při každém příjmu na sklad při nákupu z EU. 
Tvar výrobního čísla je XX*YYYYYY*ZZZZZZ…., kde XX je stát původu, YYYYYY je identifikace adresáta  výrobce a ZZZ… je případná šarže.
Pro usnadnění zadávání tohoto VČ je vytvořena doplňující vlastnost formuláře „VALIVC“ nad událostmi skladu, která je spouští automaticky pomocí masky na VČ.
Pokud není původ zboží evidován pomocí VČ, je možnost doplnění této informace až na volné kartě Intrastat.


2. Další rozšíření

2.1. Volná karta Intrastat

V ideálním případě není potřeba volnou kartu k Intrastatu používat, je to za těchto předpokladů:

· Každý artikl má u sebe uveden vztah k měrné jednotce „kg“, nebo je na události obchodu množtví nebo referenční množství uvedeno v kilogramech.

· Každému artiklu je v událostech obchodu přiřazen kód z celního sazebníku.

· Pokud je použit artikl s kódem zboží, u kterého se vykazuje DMJ, je tato hodnota uvedena v atributu množství na události obchodu, tedy například, 100 MTK v množství a 5 bal v referenčním množství.

· Pokud je na události obchodu vyplněno výrobní číslo zboží obsahující na prvních dvou pozicích stát původu zboží, třetí znak výrobního čísla musí být „*”.

Volnou kartu Intrastat není nutné vyplňovat celou, je možné kombinovat varianty vyplnění hodnot hmotnosti, kódu zboží a DMJ s údaji na události obchodu.

2.2. Předlohy výstupu Intrastat nad událostmi obchodu

Jedná se o excelovské předlohy výstupu Intrastat a Intrastat položkově, které zobrazují data vykazovaná za zadaný měsíc. 

Aktuální kurzy: http://www.cs.mfcr.cz/CmsGrc/Servis/Kurzy.htm





3. Doplňující vlastnost formuláře – Generování INSTATONLINE

Program vytváří do adresáře Intrastatonline soubor „rok_mesic.csv“, který zapakuje programem 7za. Tento zapakovaný soubor se pak natahuje web celní správy do aplikace INSTATONLINE - http://www.cs.mfcr.cz/CmsGrc/Obchod-se-zbozim/Intrastat/INSTATONLINE.htm.

Výpis příkladu programu (pro různé firmy se může program mírně lišit) :

*** GENEROVANI INTRASTATU
*** Petr Kukla - červen 2008

LOCAL _o

_o=createobject('bm_code_intrastatonline')
_o.start(_this)


****************************************************************************
* DEFINE CLASS bm_code_intrastatonline
****************************************************************************
DEFINE CLASS bm_code_intrastatonline as bm_code_sysbas
	kurs_eur=0
	kurs_gbp=0
	kurs_usd=0
	frm=.F.
	_this=''
	
PROCEDURE start(_this)
LOCAL _vysledek
this._this=_this
*_vysledek=this.dialog()
*if _vysledek=.T.
	this.main()
*endif	


*************************************************************************	

PROCEDURE main

LOCAL _file,_f,_radek, _pomocna
LOCAL _mesic,_rok,_ico,_dovoz,_zeme_op,_zeme_puvodu,_kraj,_povaha,_doprava,_dod_podminky,_zvl_pohyb,;
   _kod_zbozi,_stat_znak,_nazev_zbozi,_vlastni_hmotnost,_vlastni_hmotnost_vk,_mn_mj,_castka1,_castka2,_inter1,_inter2
   
*PRO VOLNOU KARTU
LOCAL _kod_zbozi_vk 


CREATE CURSOR data ;
(mesic C(2),;
rok C(4),;
dico C(10),;
dovoz C(1),;
zeme_op C(3),;
kraj C(2),;
zeme_puvodu C(3),;
povaha C(1),;
doprava C(1),;
dod_podminky C(3),;
zvl_pohyb C(2),;
kod_zbozi C(8),;
stat_znak C(2),;
nazev_zbozi C(80),;
vlastni_hmotnost N(20,4),;
mn_mj N(20,4),;
castka1 N(20,4),;
castka2 N(20,4),;
inter1 C(1),;
inter2 C(1) )

CREATE CURSOR data1 ;
(mesic C(2),;
rok C(4),;
dico C(10),;
dovoz C(1),;
zeme_op C(3),;
kraj C(2),;
zeme_puvodu C(3),;
povaha C(1),;
doprava C(1),;
dod_podminky C(3),;
zvl_pohyb C(2),;
kod_zbozi C(8),;
stat_znak C(2),;
nazev_zbozi C(80),;
vlastni_hmotnost C(9),;
mn_mj C(9),;
castka1 C(14),;
castka2 C(14),;
inter1 C(1),;
inter2 C(1) )


 
_file=SYS(5)+CURDIR(SYS(5))+'INTRASTATONLINE\'+SUBSTR(this._this.obec_vyber.p[2],1,4)+'_'+SUBSTR(this._this.obec_vyber.p[1],1,2)+'.csv'
 
IF FILE(_file) 
 DELETE FILE _file
ENDIF
 
_f=FCREATE(_file)
 
 _mesic   =SUBSTR(this._this.obec_vyber.p[1],1,2)
 _rok   =SUBSTR(this._this.obec_vyber.p[2],1,4)
 _dico   =tool.sql_dotaz([select ad01dico from ADADSAT where TRIM(ad01id)=(select TRIM(bw28hodn) ;
      from BWMEMOR where bw28sbr='ADM1      ' and bw28id='ad01id    ')]) 
      

SELECT OBUDAL
SCAN ALL

select data
append blank
replace mesic with _mesic, rok with _rok, dico with SUBSTR(_dico,3,10) 
select OBUDAL
 
 *_radek   =_mesic+';'+_rok+';'+SUBSTR(_dico,3,10) 
 
* PrijatéDL=3, VydanéDL=4
 _dovoz   =ob02privyd-2  
 
 if _dovoz=1
 	* NAKUP
 	* STAT ODESLANI je bran z ad01id
 	* STAT PUVODU je bran z VC udalosti (prvni 2 znaky ve VC) nebo z VK Intrastat - atribut STATPUVODU
 
	 _zeme_op=tool.sql_dotaz([SELECT ADS.ad05id FROM %%%.ADSPOJ ADP, %%%.ADADR ADR, %%%.ADSTATY ADS where ]+;
	[ ADP.AD01ID=?_p1 and ADP.AD11CISSPO=?_p2 and ADP.AD01ID=ADR.AD01ID and ADP.ad02cisadr=ADR.ad02cisadr and ]+;
	[ ADR.ad05id=ADS.ad05id], ad01id,ad11cisspo)
	 
	 if SUBSTR(OBUDAL.ob02vc,3,1)='*'
	 	_zeme_puvodu=UPPER(LEFT(OBUDAL.ob02vc,2))
	 else
	 	_zeme_puvodu=NVL(tool.sql_dotaz([select VK.STATPUVODU from BM_SERVIS.OBUDAL_INTRASTA VK where VK.ob02=?_p1], OBUDAL.ob02),'  ')
	endif

	_kraj='  '

else

 	* PRODEJ
 	* STAT URCENI je bran z ad01id1
 	* KRAJ PUVODU je bran z VC udalosti (prvni 2 znaky ve VC) nebo z VK Intrastat - atribut KRAJPUVODU
 
	 _zeme_op=tool.sql_dotaz([SELECT ADS.ad05id FROM %%%.ADSPOJ ADP, %%%.ADADR ADR, %%%.ADSTATY ADS where ]+;
	[ ADP.AD01ID=?_p1 and ADP.AD11CISSPO=?_p2 and ADP.AD01ID=ADR.AD01ID and ADP.ad02cisadr=ADR.ad02cisadr and ]+;
	[ ADR.ad05id=ADS.ad05id], ad01id1,ad11cissp1)
	 
	 _zeme_puvodu='  '

	 if SUBSTR(OBUDAL.ob02vc,3,1)='*'
	 	if UPPER(LEFT(OBUDAL.ob02vc,2))='CZ' 
	 		_kraj='31'
	 	else
	 		_kraj='99'
	 	endif
	 else
	 	_kraj=NVL(tool.sql_dotaz([select VK.KRAJPUVODU from BM_SERVIS.OBUDAL_INTRASTA VK where VK.ob02=?_p1], OBUDAL.ob02),'  ')			
	endif

endif

 
 * KRAJ

 _povaha  =SUBSTR(pa13id,1,1)+' '
 _doprava  =SUBSTR(pa13id,3,1)
 _dod_podminky =SUBSTR(pa13id,5,3)
 _zvl_pohyb  =SUBSTR(pa13id,9,2)
 
select data
replace dovoz with ALLTRIM(STR(_dovoz)), zeme_op with ALLTRIM(_zeme_op), kraj with ALLTRIM(_kraj), zeme_puvodu with ALLTRIM(_zeme_puvodu), povaha with _povaha, doprava with _doprava, ;
	dod_podminky with _dod_podminky, zvl_pohyb with _zvl_pohyb
select OBUDAL
 
 _kod_zbozi=tool.sql_dotaz([select ISNULL(ob25id,'') from OBARTIKL where sp60id=?_p1 and sp61id=?_p2],OBUDAL.sp60id,OBUDAL.sp61id)
 _kod_zbozi_vk=tool.sql_dotaz([select ISNULL(celsaz,'') from OBUDAL_INTRASTA where ob02=?_p1],OBUDAL.ob02)
 
 _kod_zbozi=MAX(_kod_zbozi, _kod_zbozi_vk)
 
  IF empty(_kod_zbozi)
   WAIT WIND 'U jedné z událostí nemá artikl vyplněn kód zboží ani v artiklu ani na vk  '+sp61id+'-'+sp60id		
   FCLOSE(_f)
   RETURN
  ENDIF
 
 
 * potrebuju zjistit jestli nepatri do skupiny, kde uvadim statisticky znak 
 *AKTUALIZOVANO pro rok 2007
 * INLIST MUZE MIT JEN 24 PARAMETRU, PROTO JE ROZDELEN NA DVA
 IF (INLIST(SUBSTR(_kod_zbozi,1,8),'22071000','22072000','27101121','27001125','27101141','27101190','27101929','27101941','27101945',;
  	'27101949','27101961','27101963','27101965','27101969','27101981','27101983','27101985','27101987','27101991') or ;
    INLIST(SUBSTR(_kod_zbozi,1,8),'27101993','27101999','27132000','27150000','29091900','34031991','34039910','38249098'))


  _stat_znak=SUBSTR(_kod_zbozi,9,2)
 ELSE
  _stat_znak=''
 ENDIF  
 
select data
replace kod_zbozi with SUBSTR(_kod_zbozi,1,8), stat_znak with _stat_znak
select OBUDAL

 
* _radek=_radek+';'+SUBSTR(_kod_zbozi,1,8)+';'+_stat_znak
 
 _nazev_zbozi=tool.sql_dotaz([select sp62artikl from SPARTIKN where sp61id=?_p1 and sp60id=?_p2 and sp38id='CZE'],;
     sp61id, sp60id)
*_nazev_zbozi=''     

select data
replace nazev_zbozi with PADR(_nazev_zbozi,80)
select OBUDAL
*_radek=_radek+';'+PADR(_nazev_zbozi,80)
 
*KILOGRAM
 _vlastni_hmotnost_vk=tool.sql_dotaz([select ISNULL(hmotnost,0) from OBUDAL_INTRASTA where ob02=?_p1],OBUDAL.ob02)

IF upper(sp24idmj1)='KGN' or upper(sp24idmj1)='KG'
  _vlastni_hmotnost=ob02mnr
ELSE
	 IF upper(sp24idmj)='KGN' or upper(sp24idmj)='KG'
	 _vlastni_hmotnost=ob02mn
 
	 ELSE
 
	  _vlastni_hmotnost=ob02mnr*tool.sql_dotaz([select ISNULL(sp59podil,0) from SPMJP where sp61id=?_p1 ]+;
	   [and sp60id=?_p2 and TRIM(upper(sp24idmj2))='KG' and sp24idmj=?_p3], sp61id, sp60id, sp24idmj1) 
	 ENDIF 
ENDIF 

_vlastni_hmotnost=MAX(_vlastni_hmotnost, _vlastni_hmotnost_vk) 
 
 IF _vlastni_hmotnost=0
   WAIT WIND 'U jedné z událostí není vyplněna hmotnost na volné kartě ani u artiklu '+sp61id+'-'+sp60id	
  FCLOSE(_f)
  RETURN
 ENDIF


* ZJISTUJI JESTLI NENI DMJ=ZZZ 
_pomocna=tool.sql_dotaz([select OBCELPRC.sp24idmj2 from OBCELPRC OBCELPRC]+;
[ where OBCELPRC.ob25id=?_p1 and  ]+ ;
[ OBCELPRC.ob26datum<=GETDATE() and OBCELPRC.ob26datumd>=GETDATE()], _kod_zbozi) 

if _pomocna='ZZZ'
	_mn_mj=0
else
	DO CASE
	CASE _pomocna='MTR' and UPPER(ALLTRIM(OBUDAL.sp24idmj))='M' 
		_mn_mj=OBUDAL.ob02mn
	CASE _pomocna='MTR' and UPPER(ALLTRIM(OBUDAL.sp24idmj1))='M' 
		_mn_mj=OBUDAL.ob02mnr
	CASE _pomocna='MTK' and UPPER(ALLTRIM(OBUDAL.sp24idmj))='M2' 
		_mn_mj=OBUDAL.ob02mn
	CASE _pomocna='MTK' and UPPER(ALLTRIM(OBUDAL.sp24idmj1))='M2' 
		_mn_mj=OBUDAL.ob02mnr
	CASE _pomocna='LTR' and UPPER(ALLTRIM(OBUDAL.sp24idmj))='L' 
		_mn_mj=OBUDAL.ob02mn
	CASE _pomocna='LTR' and UPPER(ALLTRIM(OBUDAL.sp24idmj1))='L' 
		_mn_mj=OBUDAL.ob02mnr	
	CASE _pomocna='PCE' and UPPER(ALLTRIM(OBUDAL.sp24idmj))='KS' 
		_mn_mj=OBUDAL.ob02mn
	CASE _pomocna='PCE' and UPPER(ALLTRIM(OBUDAL.sp24idmj1))='KS' 
		_mn_mj=OBUDAL.ob02mnr	
	CASE _pomocna='TNE' and UPPER(ALLTRIM(OBUDAL.sp24idmj))='T' 
		_mn_mj=OBUDAL.ob02mn
	CASE _pomocna='TNE' and UPPER(ALLTRIM(OBUDAL.sp24idmj1))='T' 
		_mn_mj=OBUDAL.ob02mnr	
	
	
	OTHERWISE
		_mn_mj=tool.sql_dotaz([select ISNULL(mndmj,0) from OBUDAL_INTRASTA where ob02=?_p1],OBUDAL.ob02)	
	ENDCASE
endif	

 IF UPPER(_pomocna)<>'ZZZ' and _mn_mj=0
   WAIT WIND 'U jedné z událostí není definováno množství v doplňkové mj  '+sp61id+'-'+sp60id		
   FCLOSE(_f)
  RETURN
 ENDIF

 
select data
replace vlastni_hmotnost with _vlastni_hmotnost, mn_mj with _mn_mj
select OBUDAL

*****************************************************************************************
 *PREPOCET DLE KURSU 
 
 LOCAL _zaklad_zm_polozka, _zaklad_zm_suma, _preprava, _castka_zm_s_dopravou, _kurscs
 
 _zaklad_zm_polozka=OBUDAL.dp02czazm
 
_zaklad_zm_suma=tool.sql_dotaz([select ISNULL(SUM(OB.dp02czazm),0) from %%%.OBUDAL OB where ]+;
[OB.sp33cisd=?_p1 and OB.sp33cisdr=?_p2 and OB.sp33cisdpm=?_p3 and ]+;
[LEFT(OB.sp61id,1) in ('M','Z') ], OBUDAL.sp33cisd, OBUDAL.sp33cisdr, OBUDAL.sp33cisdpm)

_preprava=tool.sql_dotaz([select ISNULL(SUM(OB.dp02czazm),0) from %%%.OBUDAL OB where ]+;
[OB.sp33cisd=?_p1 and OB.sp33cisdr=?_p2 and OB.sp33cisdpm=?_p3 and ]+;
[LEFT(OB.sp61id,1) not in ('M','Z') ], OBUDAL.sp33cisd, OBUDAL.sp33cisdr, OBUDAL.sp33cisdpm)

_castka_zm_s_dopravou=(_preprava*_zaklad_zm_polozka/_zaklad_zm_suma)+_zaklad_zm_polozka
 
 
 if OBUDAL.oc01id<>'CZK'
 	_kurscs=tool.sql_dotaz([select %%%.bmfce_N('KURSCS      ',NULL,?_p1,?_p2)], OBUDAL.oc01id, OBUDAL.ob02datdod)
 	if _kurscs=0
 		   WAIT WIND 'V oceňování není zadán kurs Celní správy pro toto období a měnu '+OBUDAL.oc01id		
  		   FCLOSE(_f)
		   RETURN
	endif	
	_castka1=_castka_zm_s_dopravou*_kurscs
else
	_castka1=_castka_zm_s_dopravou
endif	 
  
_castka2=0

select data
replace castka1 with _castka1, castka2 with _castka2, inter1 with ' ', inter2 with ' '
select OBUDAL
*****************************************************************************************
 
ENDSCAN

select data
index on mesic+rok+dico+dovoz+zeme_op+kraj+zeme_puvodu+povaha+doprava+dod_podminky+zvl_pohyb+kod_zbozi+stat_znak tag klic
set order to tag klic

local _klic, _aktualni_klic
local _hmot, _dmj, _c1, _c2

_klic=''
_aktualni_klic=''
_hmot=0
_dmj=0
_c1=0
_c2=0
scan
	_aktualni_klic=mesic+rok+dico+dovoz+zeme_op+kraj+zeme_puvodu+povaha+doprava+dod_podminky+zvl_pohyb+kod_zbozi+stat_znak
	if (_klic<>_aktualni_klic and !empty(_klic))
		select data1
		append blank
		gather memvar
				
		IF _hmot<1 and _hmot>0
		  _hmot=1
		ENDIF
			 
		IF _dmj<1 and _dmj>0
		  _dmj=1
		ENDIF
		
		replace vlastni_hmotnost with PADL(ALLTRIM(STR(ROUND(_hmot,0))),9,' '), mn_mj with PADL(ALLTRIM(STR(ROUND(_dmj,0))),9,' '),;
			 castka1 with PADL(ALLTRIM(STR(ROUND(_c1,0))),14,' '), castka2 with PADL(ALLTRIM(STR(ROUND(_c2,0))),14,' '), nazev_zbozi with SPACE(80)
		
		select data
		_hmot=0
		_dmj=0
		_c1=0
		_c2=0
	
	endif	

	_klic=_aktualni_klic
	_hmot=_hmot+vlastni_hmotnost
	_dmj=_dmj+mn_mj
	_c1=_c1+castka1
	_c2=_c2+castka2
	scatter memvar	

endscan

*POSLEDNI RADKY*******************************************************
		select data1
		append blank
		gather memvar
				
		IF _hmot<1 and _hmot>0
		  _hmot=1
		ENDIF
			 
		IF _dmj<1 and _dmj>0
		  _dmj=1
		ENDIF
		
		replace vlastni_hmotnost with PADL(ALLTRIM(STR(ROUND(_hmot,0))),9,' '), mn_mj with PADL(ALLTRIM(STR(ROUND(_dmj,0))),9,' '),;
			 castka1 with PADL(ALLTRIM(STR(ROUND(_c1,0))),14,' '), castka2 with PADL(ALLTRIM(STR(ROUND(_c2,0))),14,' '), nazev_zbozi with SPACE(80)
		
		select data
******************************************************************************		
		

select data1
scan
	_radek=mesic+';'+;
			rok+';'+;
			dico+';'+; 
			dovoz+';'+;
			zeme_op+';'+; 
			kraj+';'+; 
			zeme_puvodu+';'+; 
			povaha +';'+;
			doprava +';'+;
			dod_podminky+';'+;
			zvl_pohyb +';'+;
			kod_zbozi +';'+;
			stat_znak +';'+;
			nazev_zbozi +';'+;
			vlastni_hmotnost +';'+;
			mn_mj +';'+;
			castka1 +';'+;
			castka2 +';'+;
			inter1 +';'+;
			inter2+CHR(13)+CHR(10)
	 FWRITE(_f,_radek)  
endscan

FCLOSE(_f)
 

LOCAL _csv, _zip, _7za
_7za=SYS(5)+CURDIR(SYS(5))+'INTRASTATONLINE\7za.exe'
_csv=SYS(5)+CURDIR(SYS(5))+'INTRASTATONLINE\'+SUBSTR(this._this.obec_vyber.p[2],1,4)+'_'+SUBSTR(this._this.obec_vyber.p[1],1,2)+'.csv'
_zip=SYS(5)+CURDIR(SYS(5))+'INTRASTATONLINE\'+SUBSTR(this._this.obec_vyber.p[2],1,4)+'_'+SUBSTR(this._this.obec_vyber.p[1],1,2)+'.zip'

RUN /N &_7za. a -tzip &_zip. &_csv.

wait wind 'Soubor pro INTRASTATONLINE byl úspěšně vygenerován.'


ENDDEFINE
********************************************************************************
* END CLASS bm_code_intrastatonline as bm_code_sysbas
********************************************************************************



#INCLUDE SYSDIA
#INCLUDE SYSBAS

4. INSTATONLINE

INSTATONLINE - http://www.cs.mfcr.cz/CmsGrc/Obchod-se-zbozim/Intrastat/INSTATONLINE.htm

Návod k přihlášení do aplikace InstatOnline:
http://www.cs.mfcr.cz/NR/rdonlyres/482038E1-6C7F-440E-B0B4-BDE0ED82F87C/7673/Navod1.doc

